clear

*Load Data
use "./data_geo.dta", clear

*****************************************************************************
*** Setup: see Kashyap & Stein (2000) and/or Den Haan, Sumner, Yamashiro (2002) for details ***

*Drop non-states 
drop if rssd9200 == "0"
keep if  rssd9210>0 & rssd9210< 57

*Drop Delaware and/or South Dakota (see Jayaratne & Strahan 1998 JLE)
*drop if rssd9210 == 10
*drop if rssd9210 == 46

*Sample period
keep if year_q < tq(1995q1)

*Positive Assets
keep if rcfd2170 > 0

*Commercial Banks Only
keep if rssd9048 == 200 

*Insured
*keep if rssd9424 == 1 | rssd9424 == 2 | rssd9424 == 6
drop if rssd9424 == 0

*Merger
drop if MERGE_CD != .

*Set panel data
xtset rssd9001 year_q

*Fix Total Loans pre-1984 (insert before calcuating loans2 above)
gen loans2 = rcfd1400
replace loans2 = rcfd1400+rcfd2165 if year_q < tq(1984q1)

*Put loans and loan components in real terms
gen c_p_i = cpi/100
gen r_loans = loans2/c_p_i 
gen r_ci = rcfd1600/c_p_i
gen r_rlest = rcfd1410/c_p_i
gen r_con = rcfd1975/c_p_i
gen r_ag = rcfd1590/c_p_i

*Log difference 
gen loans = ln(r_loans) - ln(L.r_loans)
gen ci = ln(r_ci) - ln(L.r_ci) 
gen re = ln(r_rlest) - ln(L.r_rlest) 
gen con = ln(r_con) - ln(L.r_con)
gen ag = ln(r_ag) - ln(L.r_ag) 

*Drop outliers
sort year_q
by year_q: egen sdv = sd(loans)
drop if loans >= 5*sdv
drop if loans <= -5*sdv

*Loan Component Outliers
gen ci_share = rcfd1600/loans2
*drop if ci_share < 0.05
gen re_share = rcfd1410/loans2
*drop if re_share < 0.05
gen con_share = rcfd1975/loans2
*drop if con_share < 0.05
gen ag_share = rcfd1590/loans2
*drop if ag_share < 0.05

*Four Consecutive Quarters of Loan Growth
*ssc install tsspell
tsspell, c(loans > 0)
sort rssd9001
by rssd9001: egen maxrun = max(_seq)
drop if maxrun < 4

*Percentiles
sort year_q
by year_q: egen p95 = pctile(rcfd2170), p(95)
by year_q: egen p99 = pctile(rcfd2170), p(99)

*Generate Quarter/Year/timetrend Series
sort rssd9001 year_q
gen Quarter = quarter(date2)
gen Year = year(date2) 
egen time = group(year_q)

*Choose to keep (or drop) year deregulation was implemented 
replace intra = 1 if intra == .
replace inter =1 if inter == .

* Generate Interaction Terms
gen rr_intra = rr*intra
gen rr_intra1 = L1.rr*intra
gen rr_intra2 = L2.rr*intra
gen rr_intra3 = L3.rr*intra
gen rr_intra4 = L4.rr*intra

gen rr_inter = rr*inter
gen rr_inter1 = L1.rr*inter
gen rr_inter2 = L2.rr*inter
gen rr_inter3 = L3.rr*inter
gen rr_inter4 = L4.rr*inter

*******************************************************************************
*** Bank-level Characteristics ***

* BHC Status
gen bhc = 0
replace bhc = 1 if rssd9379 != 0

* Size (total assets)
*gen Size = 0
*replace Size = 1 if rcfd2170 >= p95
gen Size = ln(rcfd2170)

* Liquidity ratio
gen Liq2 = rcfd0010/rcfd2950
gen Sec = rcfd0390 + rcfd1350
gen sec84 = rcfd0400 + rcfd0600 + rcfd0900 + rcfd0380 + rcfd1350
replace Sec = sec84 if year_q < tq(1984q1) 
gen Liq = Sec/rcfd2170

*Equity raito
gen Cap = rcfd3210/rcfd2170

********************************************************************************
*** Loan Category Regressions ***

*** Panel (a) ***
xtreg ci_share pi_pchg hpi_chg L.Liq L.Cap L.Size L.bhc intra inter i.year_q, fe vce(cluster year_q) nonest
lincom inter

xtreg re_share pi_pchg hpi_chg L.Liq L.Cap L.Size L.bhc intra inter i.year_q, fe vce(cluster year_q) nonest
lincom inter

xtreg con_share pi_pchg hpi_chg L.Liq L.Cap L.Size L.bhc intra inter i.year_q, fe vce(cluster year_q) nonest
lincom inter


*** Panel (b) ***
*NOTE: need to drop banks with less than 5% of loans in each category before running category regressions

xtreg ci L(1/4).ci L(0/4).gdp_chg L(0/4).pce_chg L(0/4).pi_pchg L(0/4).hpi_chg L(0/4).vwretx L(0/4).rr intra inter rr_intra rr_intra1 rr_intra2 rr_intra3 rr_intra4 rr_inter rr_inter1 rr_inter2 rr_inter3 rr_inter4 i.Quarter i.Year, fe vce(cluster rssd9210) nonest
lincom rr + L1.rr + L2.rr + L3.rr + L4.rr
lincom rr_inter + rr_inter1 + rr_inter2 + rr_inter3 + rr_inter4

xtreg ci L(1/4).ci L(0/4).pi_pchg L(0/4).hpi_chg intra inter rr_intra rr_intra1 rr_intra2 rr_intra3 rr_intra4 rr_inter rr_inter1 rr_inter2 rr_inter3 rr_inter4 i.year_q, fe vce(cluster year_q) nonest
lincom rr_inter + rr_inter1 + rr_inter2 + rr_inter3 + rr_inter4

xtreg re L(1/4).re L(0/4).gdp_chg L(0/4).pce_chg L(0/4).pi_pchg L(0/4).hpi_chg L(0/4).vwretx L(0/4).rr intra inter rr_intra rr_intra1 rr_intra2 rr_intra3 rr_intra4 rr_inter rr_inter1 rr_inter2 rr_inter3 rr_inter4 i.Quarter i.Year, fe vce(cluster rssd9210) nonest
lincom rr + L1.rr + L2.rr + L3.rr + L4.rr
lincom rr_inter + rr_inter1 + rr_inter2 + rr_inter3 + rr_inter4

xtreg re L(1/4).re L(0/4).pi_pchg L(0/4).hpi_chg intra inter rr_intra rr_intra1 rr_intra2 rr_intra3 rr_intra4 rr_inter rr_inter1 rr_inter2 rr_inter3 rr_inter4 i.year_q, fe vce(cluster rssd9210) nonest
lincom rr_inter + rr_inter1 + rr_inter2 + rr_inter3 + rr_inter4

xtreg con L(1/4).con L(0/4).gdp_chg L(0/4).pce_chg L(0/4).pi_pchg L(0/4).hpi_chg L(0/4).vwretx L(0/4).rr intra inter rr_intra rr_intra1 rr_intra2 rr_intra3 rr_intra4 rr_inter rr_inter1 rr_inter2 rr_inter3 rr_inter4 i.Quarter i.Year, fe vce(cluster rssd9210) nonest
lincom rr + L1.rr + L2.rr + L3.rr + L4.rr
lincom rr_inter + rr_inter1 + rr_inter2 + rr_inter3 + rr_inter4
 
xtreg con L(1/4).con L(0/4).pi_pchg L(0/4).hpi_chg intra inter rr_intra rr_intra1 rr_intra2 rr_intra3 rr_intra4 rr_inter rr_inter1 rr_inter2 rr_inter3 rr_inter4 i.year_q, fe vce(cluster rssd9210) nonest
lincom rr_inter + rr_inter1 + rr_inter2 + rr_inter3 + rr_inter4

********************************************************************************
